clear
set more off

*************************************************
*Individual Level Regressions
*************************************************

*Get data
use ${tempdir}/panelreg_inddata_30yr.dta, clear


* Regressions of ever married status
egen firstcode = group(sex first year2)

gen int bpl2 = bpl
replace bpl2 = 9999 if bpl>100

gen byte foreign_born = bpl2>100 if bpl2~=.

gen logoccscore_gap = logoccscore_father_spouse - logoccscore_father
gen abs_agegap = abs(agegap)

gen urb = urban==2 if urban!=.
label var urb "Urban dummy"


foreach y in $yearlist  {
		gen y`y' = year2==`y'
}

local dep_ever_married = 1
local dep_logoccscore_dadinlaw_temp = 2

local inc_score = 1
local inc_1900 = 2
local inc_lido = 3
local inc_padj = 4
local inc_rfadj = 5

gen logoccscore_own = log(occscore)

foreach y in $yearlist  {
	reg logoccscore_own logoccscore_father if sex == 1 & year2 == `y'
	}
	
	
gen logoccscore_dadinlaw_temp = logoccscore_father_spouse 
gen logoccscore_husband_temp = log(husboccscore)
gen husb_parent_qtl = qtl_score_wgt_full_spouse

cap postclose results
postfile results incvar depvar spec qtile year b se using "$tempdir/reg_robust_income.dta", replace

foreach inc in score 1900 lido padj rfadj {

capture drop q1* q2* q3* q4* 

forvalues i = 1/4 {
	gen q`i' = qtl_`inc'_wgt_full==`i'
	replace q`i'=. if qtl_`inc'_wgt_full==.
}


forvalues i=1/4 {
	foreach y in $yearlist {
		gen q`i'_`y' = q`i'*y`y'
	}
}

foreach var of varlist ever_married logoccscore_dadinlaw_temp  {

* Ever married regressions -- no bpl controls
reg `var' $varlist1 i.age foreign_born if sex==2, cluster(firstcode)
gen s1_`var'`inc_`inc'' = e(sample)

forvalues i=2/4 {
	*local b`i'_1880 = _b[q`i']
	foreach y in $yearlist   {
		local b`i'_`y' = _b[q`i'_`y']
		local se`i'_`y' = _se[q`i'_`y']
	}
}

forvalues i=2/4 {
	foreach y in $yearlist   {
		post results (`inc_`inc'') (`dep_`var'') (1) (`i') (`y') (`b`i'_`y'') (`se`i'_`y'')
	}
}

* Ever married regressions -- full bpl controls
reg `var' $varlist1 i.age i.bpl2##i.year2 if sex==2, cluster(firstcode)
gen s2_`var'`inc_`inc'' = e(sample)


forvalues i=2/4 {
	*local b`i'_1880 = _b[q`i']
	foreach y in $yearlist   {
		local b`i'_`y' = _b[q`i'_`y']
		local se`i'_`y' = _se[q`i'_`y']
	}
}

forvalues i=2/4 {
	foreach y in $yearlist  {
		post results (`inc_`inc'') (`dep_`var'') (2) (`i') (`y') (`b`i'_`y'') (`se`i'_`y'')
	}
}


}
}

postclose results

save "${tempdir}reg_data_income.dta", replace 

clear 
save "${tempdir}reg_nobs_income.dta", replace empty

local dep_ever_married = 1
local dep_logoccscore_dadinlaw_temp = 2

foreach inc in 1 2 3 4 5  {
foreach var in ever_married logoccscore_dadinlaw_temp  {
foreach spec in 1 2 {

use "${tempdir}reg_data_income.dta", clear

keep if s`spec'_`var'`inc' == 1

sort year2 first
by year2 first: gen nnames_female = _n == 1

if "`var'" == "logoccscore_dadinlaw_temp" {
sort year2 spousefirst
by year2 spousefirst: gen nnames_male = _n == 1
	}
	
collapse (sum) nobs=s`spec'_`var'`inc' nnames_*, by(year2)	

gen incvar = `inc' 
gen depvar = `dep_`var''
gen spec = `spec'

rename year2 year 

append using "${tempdir}reg_nobs_income.dta"
save "${tempdir}reg_nobs_income.dta", replace 

}
}
}


use "$tempdir/reg_robust_income.dta", clear
merge m:1 incvar depvar spec year using "${tempdir}reg_nobs_income.dta"

export excel using "$outputdir/Regression Robustness.xlsx", sheet("income", replace) firstrow(var)

erase "${tempdir}reg_data_income.dta"
erase "${tempdir}reg_nobs_income.dta"








